

```

10 CLS: REV$ = "$Id: dproto.bas,v 1.2 2001-02-01 12:31:40-08 Hamilton Exp Hamilton $"
20 PRINT REV$
30 PRINT "Example D-Protocol Commands"
40 PRINT "Unsupported sample code from Pelco"
50 '
60 OPEN "COM1:4800,N,8,1" AS #2
70 CMNDCOUNT = 0
80 PANSPEED = 32
90 TILTSPEED = 32
100 SYNC = 255: REM Never changes
110 ADDRESS = 2
120 CMND1 = 0
130 CMND2 = 0
140 DATA1 = 0
150 DATA2 = 0
160 '
170 REM Get direction to move the dome
180 KPAD$ = INKEY$
190 IF LEN(KPAD$) = 0 THEN 170
200 IF KPAD$ = "x" OR KPAD$ = "X" THEN 740
210 IF KPAD$ = "s" OR KPAD$ = "S" THEN 380
220 IF KPAD$ = "a" OR KPAD$ = "A" THEN 470
230 IF LEN(KPAD$) = 1 THEN BYTE3 = ASC(KPAD$): GOTO 550
240 IF LEN(KPAD$) <> 2 THEN 170
250 BYTE3 = ASC(MID$(KPAD$,2,1))
260 '
270 REM &Hxx is used to indicate hex numbers
280 IF BYTE3 = &H4D THEN CMND2 = &H2: DATA1 = PANSPEED: ID$ = "R": GOTO 550: REM Right
290 IF BYTE3 = &H4B THEN CMND2 = &H4: DATA1 = PANSPEED: ID$ = "L": GOTO 550: REM Left
300 IF BYTE3 = &H48 THEN CMND2 = &H8: DATA2 = TILTSPEED: ID$ = "U": GOTO 550: REM Up
310 IF BYTE3 = &H50 THEN CMND2 = &H10: DATA2 = TILTSPEED: ID$ = "D": GOTO 550: REM Down
320 IF BYTE3 = &H47 THEN CMND2 = &H40: ID$ = "W": GOTO 550: REM Wide angle
330 IF BYTE3 = &H49 THEN CMND2 = &H20: ID$ = "T": GOTO 550: REM Tele
340 REM Woops it's not a used code, print it out and try again
350 PRINT """;HEX$(BYTE3);"";
360 GOTO 170
370 '
380 REM Ask for a new speed
390 INPUT "Enter in a new positive speed less than 64 (cr = 32)";SPEED
400 IF SPEED = 0 THEN SPEED = 32
410 IF SPEED > 64 OR SPEED < 0 THEN 390
420 PANSPEED = SPEED
430 TILTSPEED = SPEED: IF SPEED = 64 THEN TILTSPEED = 63
440 PRINT "Using pan/tilt speeds of: ";PANSPEED;"/";TILTSPEED
450 GOTO 170
460 '
470 REM Ask for a new camera address
480 REM Note that camera addresses are one based
490 INPUT "What camera address should be used (cr = camera 2)";ADDRESS
500 IF ADDRESS = 0 THEN ADDRESS = 2
510 IF ADDRESS > 255 OR ADDRESS < 0 THEN 490
520 PRINT "Using a camera address of: ";ADDRESS
530 GOTO 170
540 '
550 REM Send it out
560 PRINT USING"\\ ##: ";ID$,CMNDCOUNT;
570 CMNDCOUNT = CMNDCOUNT + 1: IF CMNDCOUNT > 99 THEN CMNDCOUNT = 0
580 GOSUB 890: REM Send it

```

```

590 '
600 REM For next time start with everything at zero
610 CMND1 = 0
620 CMND2 = 0
630 DATA1 = 0
640 DATA2 = 0
650 '
660 REM Now stop it moving after a second, or more, of motion
670 REM Note that the subroutines may move if this gets renumbered and
680 REM the next few lines are commented out.
690 REM GOSUB 770: REM Delay a sec
700 REM GOSUB 830: REM Send it
710 REM GOSUB 770: REM Delay a sec
720 GOTO 170
730 '
740 REM All done
750 PRINT "All done"
760 PRINT DATE$,TIME$
770 END
780 '
790 REM Wait for up to one second
800 X$ = RIGHT$(TIME$,2)
810 IF X$ = RIGHT$(TIME$,2) GOTO 810
820 RETURN
830 '
840 REM Get the checksum
850 CHECKSUM = ADDRESS + CMND1 + CMND2 + DATA1 + DATA2
860 CHECKSUM = CHECKSUM AND 255
870 RETURN
880 '
890 REM Send it out and display it
900 GOSUB 840: REM Get checksum
910 D1$ = CHR$(SYNC)
920 D2$ = CHR$(ADDRESS)
930 D3$ = CHR$(CMND1)
940 D4$ = CHR$(CMND2)
950 D5$ = CHR$(DATA1)
960 D6$ = CHR$(DATA2)
970 D7$ = CHR$(CHECKSUM)
980 PRINT #2,D1$;D2$;D3$;D4$;D5$;D6$;D7$;
990 X = SYNC: GOSUB 1090
1000 X = ADDRESS: GOSUB 1090
1010 X = CMND1: GOSUB 1090
1020 X = CMND2: GOSUB 1090
1030 X = DATA1: GOSUB 1090
1040 X = DATA2: GOSUB 1090
1050 X = CHECKSUM: GOSUB 1090
1060 PRINT
1070 RETURN
1080 '
1090 REM Print out a command byte
1100 X$ = HEX$(X)
1110 IF LEN(X$) = 1 THEN X$ = "0" + X$
1120 PRINT X$;" ";
1130 RETURN

```